In this Maple file, we compute the evolution equations for the Painlevé 2 equations using the compatibility equation of the Lax system. We also obtain the expression of the Lax matrices in the geometric gauge without apparent singularities.
The operator is \hbar (alpha111\partial_{t_{1^{(1)},1} +alpha121\partial_{t_{1^{(2)},1}))

Lax matrices in the oper gauge from previous Maple files

Summary of previous files: We have the expression for some coefficients of the Lax matrix L and of A.

```
> restart:
  with(LinearAlgebra):
  P011 := t010+t020;
  P111 := t110+t120;
  P121 := t111+t121;
  P142 := t111*t121;
  P132 := t110*t121+t111*t120;
  P022 := t010*t020;
  CoherenceEquation1:=tinfty10+tinfty20+t010+t020+t110+t120;
  P1 := unapply(P011/x+P121/(x-1)^2+P111/(x-1),x);
  P2 := unapply(P022/x^2+P012/x+P142/(x-1)^4+P132/(x-1)^3+P122/
  (x-1)^2+P112/(x-1),x);
  tdP2:=unapply(P2(lambda)-P012/lambda-P112/(lambda-1)-P122/
  (lambda-1)^2,lambda);

  c1:=(alpha111*t121-alpha121*t111)/(t111-t121);
  mu:=(alpha111-alpha121)*q*(q-1)^2/(t111-t121);
  nu0:=(alpha111-alpha121)*(q-1)^2/(t111-t121);
  nuMinus1:=(alpha111-alpha121)*(q-1)/(t111-t121);

  C12:=-a2:
  C11:=-a1:
  C01:=-a0:
  c11:=(alpha111*t121-alpha121*t111)/(t111-t121);

  dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
  dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):
  L:=Matrix(2,2,0):
  L[1,1]:=0:
  L[1,2]:=1:
  L[2,1]:=-P2(lambda)+(P012-a0)/lambda+(P122-a2)/(lambda-1)^2+
  (P112-a1)/(lambda-1) -h*p/(lambda-q):
  L[2,2]:= P1(lambda)-h/lambda-2*h/(lambda-1)+h/(lambda-q):

  Eq1:=a0+a1+h*p;
```

```
Eq2:= h*p*q+t010*t020+a1+a2-(h+tinfty20)*tinfty10;

c0:=0:
A:=Matrix(2,2,0):
A[1,1]:= c1/(lambda-1) +c0+rho/(lambda-q):
A[1,2]:= nuMinus1*lambda+nu0+mu/(lambda-q):
A[2,1]:= AA21(lambda):
A[2,2]:= AA22(lambda):
dAdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dAdlambda[i,j]:=diff
(A[i,j],lambda):  od: od:

L;
A;
P012 := -P112;
P122 := -t010*t020+tinfty10*tinfty20-P112;

Q2:=unapply(-p*(q-0)*(q-1)^2,lambda):
J:=Matrix(2,2,0):
J[1,1]:=1:
J[1,2]:=0:
J[2,1]:=Q2(lambda)/(lambda-q):
J[2,2]:=(lambda-0)^1*(lambda-1)^2/(lambda-q):
dJdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dJdlambda[i,j]:=diff
(J[i,j],lambda): od: od:
J:

LJ:=Matrix(2,2,0):
LJ[1,1]:=0:
LJ[1,2]:=0:
LJ[2,2]:=diff(J[2,2],q)*Lq+diff(J[2,2],p)*Lp+h*alpha111*diff(J
[2,2],t111)+h*alpha121*diff(J[2,2],t121):
LJ[2,1]:=diff(J[2,1],q)*Lq+diff(J[2,1],p)*Lp+h*alpha111*diff(J
[2,1],t111)+h*alpha121*diff(J[2,1],t121):
LJ:

checkL:=simplify(Multiply(Multiply(J,L),J^(-1))+h*Multiply
(dJdlambda,J^(-1))):
checkA:=simplify(Multiply(Multiply(J,A),J^(-1))+Multiply(LJ,J^
(-1))):
```

$$P011 := t010 + t020 \quad \textbf{(1.1)}$$
$$P111 := t110 + t120$$
$$P121 := t111 + t121$$
$$P142 := t111\, t121$$
$$P132 := t110\, t121 + t111\, t120$$
$$P022 := t010\, t020$$
$$CoherenceEquation1 := tinfty10 + tinfty20 + t010 + t020 + t110 + t120$$
$$P1 := x \to \frac{t010 + t020}{x} + \frac{t111 + t121}{(x-1)^2} + \frac{t110 + t120}{x-1}$$
$$P2 := x \to \frac{t010\, t020}{x^2} + \frac{P012}{x} + \frac{t111\, t121}{(x-1)^4} + \frac{t110\, t121 + t111\, t120}{(x-1)^3} + \frac{P122}{(x-1)^2} + \frac{P112}{x-1}$$
$$tdP2 := \lambda \to \frac{t010\, t020}{\lambda^2} + \frac{t111\, t121}{(\lambda-1)^4} + \frac{t110\, t121 + t111\, t120}{(\lambda-1)^3}$$
$$c1 := \frac{-t111\, \alpha l21 + t121\, \alpha l11}{t111 - t121}$$
$$\mu := \frac{(\alpha l11 - \alpha l21)\, q\, (q-1)^2}{t111 - t121}$$
$$\nu 0 := \frac{(\alpha l11 - \alpha l21)\, (q-1)^2}{t111 - t121}$$
$$nuMinus1 := \frac{(\alpha l11 - \alpha l21)\, (q-1)}{t111 - t121}$$
$$c11 := \frac{-t111\, \alpha l21 + t121\, \alpha l11}{t111 - t121}$$
$$Eq1 := h\, p + a0 + a1$$
$$Eq2 := h\, p\, q + t010\, t020 + a1 + a2 - (h + tinfty20)\, tinfty10$$
$$\left[[0, 1], \left[-\frac{t010\, t020}{\lambda^2} - \frac{P012}{\lambda} - \frac{t111\, t121}{(\lambda-1)^4} - \frac{t110\, t121 + t111\, t120}{(\lambda-1)^3} - \frac{P122}{(\lambda-1)^2} - \frac{P112}{\lambda-1} + \frac{P012 - a0}{\lambda} + \frac{P122 - a2}{(\lambda-1)^2} + \frac{P112 - a1}{\lambda-1} - \frac{h\, p}{\lambda - q}, \frac{t010 + t020}{\lambda} + \frac{t111 + t121}{(\lambda-1)^2} + \frac{t110 + t120}{\lambda-1} - \frac{h}{\lambda} - \frac{2\, h}{\lambda-1} + \frac{h}{\lambda-q} \right] \right]$$
$$\left[\left[\frac{-t111\, \alpha l21 + t121\, \alpha l11}{(t111 - t121)\, (\lambda-1)} + \frac{\rho}{\lambda-q}, \frac{(\alpha l11 - \alpha l21)\, (q-1)\, \lambda}{t111 - t121} + \frac{(\alpha l11 - \alpha l21)\, (q-1)^2}{t111 - t121} + \frac{(\alpha l11 - \alpha l21)\, q\, (q-1)^2}{(t111 - t121)\, (\lambda-q)} \right], [AA21(\lambda), AA22(\lambda)] \right]$$
$$P012 := -P112$$
$$P122 := -t010\, t020 + tinfty10\, tinfty20 - P112$$

Solving the compatibility equations to obtain the Hamiltonian evolutions.

The compatibility equation is \mathcal{L}L=h\partial_\lambda A+[A,L]
Since the first line of L is trivial, we may easily obtain A[2,1] et A[2,2] to obtain the full expression for A

```
> LL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):

  Entry11:=LL[1,1]:
  Entry12:=LL[1,2]:

  AA21:=unapply(solve(Entry11=0,AA21(lambda)),lambda):
  AA21bis:=h*dAdlambda[1,1]+A[1,2]*L[2,1]:
  simplify(AA21(lambda)-AA21bis);


  AA22:=unapply(solve(Entry12=0,AA22(lambda)),lambda):
  AA22bis:=h*dAdlambda[1,2]+A[1,1]+A[1,2]*L[2,2]:
  simplify(AA22(lambda)-AA22bis);

  simplify(Entry11);
  simplify(Entry12);
  LL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
```

$$0 \tag{2.1}$$
$$0$$
$$0$$
$$0$$

We now compute the action of \mathcal{L} on L[2,2] et L[2,1] to obtain the evolution equations

Evolution of entry L_{2,2}

```
> Entry22:=simplify(LL[2,2]);
  Entry22TermLambdaMinusqCube:=factor(residue(Entry22*(lambda-q)
  ^2,lambda=q));
  Entry22TermLambdaMinusqSquare:=factor(residue(Entry22*(lambda-
  q),lambda=q));
  Entry22TermLambdaMinusq:=factor(residue(Entry22,lambda=q));

  Entry22TermLambdaInfty1:=factor(-residue(Entry22/lambda^2,
  lambda=infinity));
  Entry22TermLambdaInfty0:=factor(-residue(Entry22/lambda,lambda=
  infinity));
```

```
Entry22TermLambda1OrderMinus1:=factor(residue(Entry22/
(lambda-1)^0,lambda=1));
Entry22TermLambda1OrderMinus2:=factor(residue
(Entry22*(lambda-1),lambda=1));
Entry22TermLambda1OrderMinus3:=factor(residue(Entry22*
(lambda-1)^2,lambda=1));

Entry22TermLambda0OrderMinus1:=factor(residue(Entry22/(lambda)
^0,lambda=0));
Entry22TermLambda0OrderMinus2:=factor(residue(Entry22*(lambda),
lambda=0));
Entry22TermLambda0OrderMinus3:=factor(residue(Entry22*(lambda)
^2,lambda=0));


simplify( Entry22-(Entry22TermLambdaMinusqSquare/(lambda-q)^2+
Entry22TermLambdaMinusq/(lambda-q)
+Entry22TermLambdaInfty0+Entry22TermLambda1OrderMinus2/
(lambda-1)^2) );
L[2,2];
```

$$Entry22 := \frac{1}{(\lambda-1)^2\,(\lambda-q)^2\,(t111-t121)}\Big(h\Big(\big((al11-al21)\,(h-t010-t020-t110-t120)\,\lambda^2-2\,(al11-al21)\,(h-t010-t020-t110-t120)\,\lambda+(h-t010-t020-t110+t111-t120-t121)\,al11-al21\,(h-t010-t020-t110-t111-t120+t121)\big)\,q^2+\big(-(al11-al21)\,(h-2\,t010-2\,t020-t110+t111-t120+t121)\,\lambda^2+\big((2\,h-4\,t010-4\,t020-2\,t110-2\,t120+4\,t121)\,al11-2\,al21\,(h-2\,t010-2\,t020-t110+2\,t111-t120)\big)\,\lambda-(al11-al21)\,(h-2\,t010-2\,t020-t110+t111-t120+t121)\big)\,q+\big((-t010-t020+t111-t121)\,al11+(t010+t020+t111-t121)\,al21-2\,\rho\,(t111-t121)\big)\,\lambda^2+\big((2\,t010+2\,t020)\,al11+(-2\,t010-2\,t020)\,al21+4\,\rho\,(t111-t121)\big)\,\lambda+(-t010-t020)\,al11+(t010+t020)\,al21-2\,\rho\,(t111-t121)\Big)\Big) \tag{2.2}$$

$$Entry22TermLambdaMinusqCube := 0$$

$$Entry22TermLambdaMinusqSquare := \frac{1}{t111-t121}\Big(\big(h\,q^2\,al11-h\,q^2\,al21-q^2\,t010\,al11+q^2\,t010\,al21-q^2\,t020\,al11+q^2\,t020\,al21-q^2\,t110\,al11+q^2\,t110\,al21-q^2\,t120\,al11+q^2\,t120\,al21-h\,q\,al11+h\,q\,al21+2\,q\,t010\,al11-2\,q\,t010\,al21+2\,q\,t020\,al11-2\,q\,t020\,al21+q\,t110\,al11-q\,t110\,al21-q\,t111\,al11+q\,t111\,al21+q\,t120\,al11-q\,t120\,al21-q\,t121\,al11$$

$$+ q\, t121\, \alpha 121 - 2\rho\, t111 + 2\rho\, t121 - t010\, \alpha 11 + t010\, \alpha 121 - t020\, \alpha 11 + t020\, \alpha 121)\, h)$$

$$Entry22TermLambdaMinusq := 0$$
$$Entry22TermLambdaInfty1 := 0$$
$$Entry22TermLambdaInfty0 := 0$$
$$Entry22TermLambda1OrderMinus1 := 0$$
$$Entry22TermLambda1OrderMinus2 := (\alpha 111 + \alpha 121)\, h$$
$$Entry22TermLambda1OrderMinus3 := 0$$
$$Entry22TermLambda0OrderMinus1 := 0$$
$$Entry22TermLambda0OrderMinus2 := 0$$
$$Entry22TermLambda0OrderMinus3 := 0$$
$$0$$
$$\frac{t010 + t020}{\lambda} + \frac{t111 + t121}{(\lambda - 1)^2} + \frac{t110 + t120}{\lambda - 1} - \frac{h}{\lambda} - \frac{2\, h}{\lambda - 1} + \frac{h}{\lambda - q}$$

Since the deformation operator is \hbar (alpha111\partial_{t_{1^{(1)},1} +alpha121\partial_{t_{1^{(2)},1}) it acts on the double pole and gives h*(alpha111+alpha121).
On the simple pole at lambda=q we find h*L[q] that should correspond to the double pole of the compatibility equation.

```
> Equation1:=Entry22TermLambda1OrderMinus2- h*(alpha111+alpha121)
  ;
```

$$Equation1 := 0 \tag{2.3}$$

```
> Lq:=factor(Entry22TermLambdaMinusqSquare/h):
  Lqbis:=-2*rho+mu/q/(q-1)^2*( (h-t120-t010-t020-t110)*q^2-q*
  (h-2*t010-2*t020-t110+t111-t120+t121)-(t010+t020));
  factor(simplify(Lq-Lqbis));
```

$$Lqbis := -2\rho + \frac{1}{t111 - t121}\left((\alpha 111 - \alpha 121)\left((h - t010 - t020 - t110 - t120)\, q^2 - q\, (h - 2\, t010 - 2\, t020 - t110 + t111 - t120 + t121) - t010 - t020\right)\right) \tag{2.4}$$
$$0$$

Evolution of \mathcal{L}[L[2,1]]

```
> Entry21:=simplify(LL[2,1]):
  Entry21TermLambdaMinusqCube:=factor(residue(Entry21*(lambda-q)
  ^2,lambda=q));
  Entry21TermLambdaMinusqSquare:=factor(residue(Entry21*(lambda-
  q),lambda=q));
  Entry21TermLambdaMinusq:=factor(residue(Entry21,lambda=q));

  Entry21TermLambdaInfty1:=factor(-residue(Entry21/lambda^2,
  lambda=infinity));
  Entry21TermLambdaInfty0:=factor(-residue(Entry21/lambda,lambda=
  infinity));

  Entry21TermLambdaZero2:=factor(residue(Entry21*(lambda),lambda=
  0));
```

```
Entry21TermLambdaZero1:=factor(residue(Entry21,lambda=0));

Entry21TermLambdaUn5:=factor(residue(Entry21*(lambda-1)^4,
lambda=1));
Entry21TermLambdaUn4:=factor(residue(Entry21*(lambda-1)^3,
lambda=1));
Entry21TermLambdaUn3:=factor(residue(Entry21*(lambda-1)^2,
lambda=1));
Entry21TermLambdaUn2:=factor(residue(Entry21*(lambda-1),lambda=
1));
Entry21TermLambdaUn1:=factor(residue(Entry21,lambda=1));

simplify( Entry21-(Entry21TermLambdaMinusqCube/(lambda-q)^3+
Entry21TermLambdaMinusqSquare/(lambda-q)^2+
Entry21TermLambdaMinusq/(lambda-q)
+Entry21TermLambdaInfty0+Entry21TermLambdaInfty1*lambda
+Entry21TermLambdaZero1/lambda+Entry21TermLambdaUn1/(lambda-1)
) );
L[2,1];
```

$$Entry21TermLambdaMinusqCube := \frac{1}{t111 - t121} \left(3 \left(p\, q^3\, \alpha 11 - p\, q^3\, \alpha 21 - 2\, p\, q^2\, \alpha 11 + 2\, p\, q^2\, \alpha 21 + p\, q\, \alpha 11 - p\, q\, \alpha 21 + \rho\, t111 - \rho\, t121\right) h^2\right) \tag{2.5}$$

$$Entry21TermLambdaMinusqSquare := \frac{1}{(q-1)^2\, (t111 - t121)\, q} \Big(h \big(2\, h\, p\, q^5\, \alpha 11 - 2\, h\, p\, q^5\, \alpha 21 + 2\, a0\, q^5\, \alpha 11 - 2\, a0\, q^5\, \alpha 21 + 2\, a1\, q^5\, \alpha 11 - 2\, a1\, q^5\, \alpha 21 - 7\, h\, p\, q^4\, \alpha 11 + 7\, h\, p\, q^4\, \alpha 21 + 2\, q^4\, t010\, t020\, \alpha 11 - 2\, q^4\, t010\, t020\, \alpha 21 - 8\, a0\, q^4\, \alpha 11 + 8\, a0\, q^4\, \alpha 21 - 6\, a1\, q^4\, \alpha 11 + 6\, a1\, q^4\, \alpha 21 + 2\, a2\, q^4\, \alpha 11 - 2\, a2\, q^4\, \alpha 21 + 9\, h\, p\, q^3\, \alpha 11 - 9\, h\, p\, q^3\, \alpha 21 - 8\, q^3\, t010\, t020\, \alpha 11 + 8\, q^3\, t010\, t020\, \alpha 21 + 2\, q^3\, t110\, t121\, \alpha 11 - 2\, q^3\, t110\, t121\, \alpha 21 + 2\, q^3\, t111\, t120\, \alpha 11 - 2\, q^3\, t111\, t120\, \alpha 21 + 12\, a0\, q^3\, \alpha 11 - 12\, a0\, q^3\, \alpha 21 + 6\, a1\, q^3\, \alpha 11 - 6\, a1\, q^3\, \alpha 21 - 4\, a2\, q^3\, \alpha 11 + 4\, a2\, q^3\, \alpha 21 - 5\, h\, p\, q^2\, \alpha 11 + 5\, h\, p\, q^2\, \alpha 21 - 3\, h\, q^2\, \rho\, t111 + 3\, h\, q^2\, \rho\, t121 + q^2\, \rho\, t010\, t111 - q^2\, \rho\, t010\, t121 + q^2\, \rho\, t020\, t111 - q^2\, \rho\, t020\, t121 + q^2\, \rho\, t110\, t111 - q^2\, \rho\, t110\, t121 + q^2\, \rho\, t111\, t120 - q^2\, \rho\, t120\, t121 + 12\, q^2\, t010\, t020\, \alpha 11 - 12\, q^2\, t010\, t020\, \alpha 21 - 2\, q^2\, t110\, t121\, \alpha 11 + 2\, q^2\, t110\, t121\, \alpha 21 - 2\, q^2\, t111\, t120\, \alpha 11 + 2\, q^2\, t111\, t120\, \alpha 21 + 2\, q^2\, t111\, t121\, \alpha 11 - 2\, q^2\, t111\, t121\, \alpha 21 - 8\, a0\, q^2\, \alpha 11 + 8\, a0\, q^2\, \alpha 21 - 2\, a1\, q^2\, \alpha 11 + 2\, a1\, q^2\, \alpha 21 + 2\, a2\, q^2\, \alpha 11 - 2\, a2\, q^2\, \alpha 21 + h\, p\, q\, \alpha 11 - h\, p\, q\, \alpha 21 + 4\, h\, q\, \rho\, t111 - 4\, h\, q\, \rho\, t121 - 2\, q\, \rho\, t010\, t111 + 2\, q\, \rho\, t010\, t121 - 2\, q\, \rho\, t020\, t111 + 2\, q\, \rho\, t020\, t121 - q\, \rho\, t110\, t111 + q\, \rho\, t110\, t121 + q\, \rho\, t111^2 - q\, \rho\, t111\, t120 + q\, \rho\, t120\, t121 - q\, \rho\, t121^2 - 8\, q\, t010\, t020\, \alpha 11 + 8\, q\, t010\, t020\, \alpha 21$$

$$+ 2\, a0\, q\, \alpha111 - 2\, a0\, q\, \alpha121 - h\, \rho\, t111 + h\, \rho\, t121 + \rho\, t010\, t111 - \rho\, t010\, t121$$
$$+ \rho\, t020\, t111 - \rho\, t020\, t121 + 2\, t010\, t020\, \alpha111 - 2\, t010\, t020\, \alpha121))$$

$$Entry21TermLambdaMinusq := -\frac{1}{(q-1)^3\, (t111 - t121)\, q^2} ((h\, p\, q^6\, \alpha111 - h\, p\, q^6\, \alpha121$$
$$+ a0\, q^6\, \alpha111 - a0\, q^6\, \alpha121 + a1\, q^6\, \alpha111 - a1\, q^6\, \alpha121 - 4\, h\, p\, q^5\, \alpha111$$
$$+ 4\, h\, p\, q^5\, \alpha121 + 2\, q^5\, t010\, t020\, \alpha111 - 2\, q^5\, t010\, t020\, \alpha121 - 5\, a0\, q^5\, \alpha111$$
$$+ 5\, a0\, q^5\, \alpha121 - 3\, a1\, q^5\, \alpha111 + 3\, a1\, q^5\, \alpha121 + 2\, a2\, q^5\, \alpha111 - 2\, a2\, q^5\, \alpha121$$
$$+ 6\, h\, p\, q^4\, \alpha111 - 6\, h\, p\, q^4\, \alpha121 - 10\, q^4\, t010\, t020\, \alpha111 + 10\, q^4\, t010\, t020\, \alpha121$$
$$+ 3\, q^4\, t110\, t121\, \alpha111 - 3\, q^4\, t110\, t121\, \alpha121 + 3\, q^4\, t111\, t120\, \alpha111$$
$$- 3\, q^4\, t111\, t120\, \alpha121 + 10\, a0\, q^4\, \alpha111 - 10\, a0\, q^4\, \alpha121 + 3\, a1\, q^4\, \alpha111 - 3\, a1\, q^4\, \alpha121$$
$$- 4\, a2\, q^4\, \alpha111 + 4\, a2\, q^4\, \alpha121 - 4\, h\, p\, q^3\, \alpha111 + 4\, h\, p\, q^3\, \alpha121 - 3\, h\, q^3\, \rho\, t111$$
$$+ 3\, h\, q^3\, \rho\, t121 + h\, q^3\, t111\, \alpha121 - h\, q^3\, t121\, \alpha111 + q^3\, \rho\, t010\, t111 - q^3\, \rho\, t010\, t121$$
$$+ q^3\, \rho\, t020\, t111 - q^3\, \rho\, t020\, t121 + q^3\, \rho\, t110\, t111 - q^3\, \rho\, t110\, t121 + q^3\, \rho\, t111\, t120$$
$$- q^3\, \rho\, t120\, t121 + 20\, q^3\, t010\, t020\, \alpha111 - 20\, q^3\, t010\, t020\, \alpha121 - 3\, q^3\, t110\, t121\, \alpha111$$
$$+ 3\, q^3\, t110\, t121\, \alpha121 - 3\, q^3\, t111\, t120\, \alpha111 + 3\, q^3\, t111\, t120\, \alpha121$$
$$+ 4\, q^3\, t111\, t121\, \alpha111 - 4\, q^3\, t111\, t121\, \alpha121 - 10\, a0\, q^3\, \alpha111 + 10\, a0\, q^3\, \alpha121$$
$$- a1\, q^3\, \alpha111 + a1\, q^3\, \alpha121 + 2\, a2\, q^3\, \alpha111 - 2\, a2\, q^3\, \alpha121 + h\, p\, q^2\, \alpha111$$
$$- h\, p\, q^2\, \alpha121 + 5\, h\, q^2\, \rho\, t111 - 5\, h\, q^2\, \rho\, t121 - h\, q^2\, t111\, \alpha121 + h\, q^2\, t121\, \alpha111$$
$$- 3\, q^2\, \rho\, t010\, t111 + 3\, q^2\, \rho\, t010\, t121 - 3\, q^2\, \rho\, t020\, t111 + 3\, q^2\, \rho\, t020\, t121$$
$$- q^2\, \rho\, t110\, t111 + q^2\, \rho\, t110\, t121 + 2\, q^2\, \rho\, t111^2 - q^2\, \rho\, t111\, t120 + q^2\, \rho\, t120\, t121$$
$$- 2\, q^2\, \rho\, t121^2 - 20\, q^2\, t010\, t020\, \alpha111 + 20\, q^2\, t010\, t020\, \alpha121 + 5\, a0\, q^2\, \alpha111$$
$$- 5\, a0\, q^2\, \alpha121 - 3\, h\, q\, \rho\, t111 + 3\, h\, q\, \rho\, t121 + 3\, q\, \rho\, t010\, t111 - 3\, q\, \rho\, t010\, t121$$
$$+ 3\, q\, \rho\, t020\, t111 - 3\, q\, \rho\, t020\, t121 + 10\, q\, t010\, t020\, \alpha111 - 10\, q\, t010\, t020\, \alpha121$$
$$- a0\, q\, \alpha111 + a0\, q\, \alpha121 + h\, \rho\, t111 - h\, \rho\, t121 - \rho\, t010\, t111 + \rho\, t010\, t121$$
$$- \rho\, t020\, t111 + \rho\, t020\, t121 - 2\, t010\, t020\, \alpha111 + 2\, t010\, t020\, \alpha121)\, h)$$

$$Entry21TermLambdaInfty1 := 0$$
$$Entry21TermLambdaInfty0 := 0$$
$$Entry21TermLambdaZero2 := 0$$

$$Entry21TermLambdaZero1 := -\frac{1}{q^2\, (t111 - t121)} (h\, (-h\, q^2\, t111\, \alpha121 + h\, q^2\, t121\, \alpha111$$
$$- 2\, q^2\, t010\, t020\, \alpha111 + 2\, q^2\, t010\, t020\, \alpha121 + q^2\, t010\, t111\, \alpha121 - q^2\, t010\, t121\, \alpha111$$
$$+ q^2\, t020\, t111\, \alpha121 - q^2\, t020\, t121\, \alpha111 + a0\, q^2\, \alpha111 - a0\, q^2\, \alpha121$$
$$+ 4\, q\, t010\, t020\, \alpha111 - 4\, q\, t010\, t020\, \alpha121 - a0\, q\, \alpha111 + a0\, q\, \alpha121 + h\, \rho\, t111$$
$$- h\, \rho\, t121 - \rho\, t010\, t111 + \rho\, t010\, t121 - \rho\, t020\, t111 + \rho\, t020\, t121 - 2\, t010\, t020\, \alpha111$$
$$+ 2\, t010\, t020\, \alpha121))$$

$$Entry21TermLambdaUn5 := 0$$
$$Entry21TermLambdaUn4 := -(t111\, \alpha121 + t121\, \alpha111)\, h$$
$$Entry21TermLambdaUn3 := -(t110\, \alpha121 + t120\, \alpha111)\, h$$

$$Entry21TermLambdaUn2 := \frac{1}{(q-1)^2\, (t111 - t121)} ((h\, q^2\, t111\, \alpha121 - h\, q^2\, t121\, \alpha111$$
$$- q^2\, t010\, t111\, \alpha121 + q^2\, t010\, t121\, \alpha111 - q^2\, t020\, t111\, \alpha121 + q^2\, t020\, t121\, \alpha111$$

$$+ q^2\, t110\, t121\, \alpha111 - q^2\, t110\, t121\, \alpha121 + q^2\, t111\, t120\, \alpha111 - q^2\, t111\, t120\, \alpha121$$
$$- h\, q\, t111\, \alpha121 + h\, q\, t121\, \alpha111 + 2\, q\, t010\, t111\, \alpha121 - 2\, q\, t010\, t121\, \alpha111$$
$$+ 2\, q\, t020\, t111\, \alpha121 - 2\, q\, t020\, t121\, \alpha111 - q\, t110\, t121\, \alpha111 + q\, t110\, t121\, \alpha121$$
$$- q\, t111\, t120\, \alpha111 + q\, t111\, t120\, \alpha121 + 2\, q\, t111\, t121\, \alpha111 - 2\, q\, t111\, t121\, \alpha121$$
$$+ \rho\, t111^2 - \rho\, t121^2 - t010\, t111\, \alpha121 + t010\, t121\, \alpha111 - t020\, t111\, \alpha121$$
$$+ t020\, t121\, \alpha111 \big)\, h \big)$$

$$Entry21TermLambdaUn1 := \frac{1}{(q-1)^3\, (t111 - t121)} \Big(\big(-h\, q^3\, t111\, \alpha121 + h\, q^3\, t121\, \alpha111$$
$$+ q^3\, t010\, t111\, \alpha121 - q^3\, t010\, t121\, \alpha111 + q^3\, t020\, t111\, \alpha121 - q^3\, t020\, t121\, \alpha111$$
$$+ a1\, q^3\, \alpha111 - a1\, q^3\, \alpha121 + 2\, a2\, q^3\, \alpha111 - 2\, a2\, q^3\, \alpha121 + 3\, h\, q^2\, t111\, \alpha121$$
$$- 3\, h\, q^2\, t121\, \alpha111 - 3\, q^2\, t010\, t111\, \alpha121 + 3\, q^2\, t010\, t121\, \alpha111 - 3\, q^2\, t020\, t111\, \alpha121$$
$$+ 3\, q^2\, t020\, t121\, \alpha111 + 3\, q^2\, t110\, t121\, \alpha111 - 3\, q^2\, t110\, t121\, \alpha121$$
$$+ 3\, q^2\, t111\, t120\, \alpha111 - 3\, q^2\, t111\, t120\, \alpha121 - 3\, a1\, q^2\, \alpha111 + 3\, a1\, q^2\, \alpha121$$
$$- 4\, a2\, q^2\, \alpha111 + 4\, a2\, q^2\, \alpha121 - 2\, h\, q\, \rho\, t111 + 2\, h\, q\, \rho\, t121 - 2\, h\, q\, t111\, \alpha121$$
$$+ 2\, h\, q\, t121\, \alpha111 + q\, \rho\, t110\, t111 - q\, \rho\, t110\, t121 + q\, \rho\, t111\, t120 - q\, \rho\, t120\, t121$$
$$+ 3\, q\, t010\, t111\, \alpha121 - 3\, q\, t010\, t121\, \alpha111 + 3\, q\, t020\, t111\, \alpha121 - 3\, q\, t020\, t121\, \alpha111$$
$$- 3\, q\, t110\, t121\, \alpha111 + 3\, q\, t110\, t121\, \alpha121 - 3\, q\, t111\, t120\, \alpha111 + 3\, q\, t111\, t120\, \alpha121$$
$$+ 4\, q\, t111\, t121\, \alpha111 - 4\, q\, t111\, t121\, \alpha121 + 3\, a1\, q\, \alpha111 - 3\, a1\, q\, \alpha121$$
$$+ 2\, a2\, q\, \alpha111 - 2\, a2\, q\, \alpha121 + 2\, h\, \rho\, t111 - 2\, h\, \rho\, t121 - \rho\, t110\, t111 + \rho\, t110\, t121$$
$$+ 2\, \rho\, t111^2 - \rho\, t111\, t120 + \rho\, t120\, t121 - 2\, \rho\, t121^2 - t010\, t111\, \alpha121 + t010\, t121\, \alpha111$$
$$- t020\, t111\, \alpha121 + t020\, t121\, \alpha111 - a1\, \alpha111 + a1\, \alpha121 \big)\, h \Big)$$
$$- \frac{1}{(q-1)^2\, (t111 - t121)\, (\lambda - 1)^4} \Bigg(h \Bigg(\Bigg(\Big(\big((-t120\, \alpha111 - \alpha121\, (h - t010 - t020$$
$$- t120))\, t111 + ((h - t010 - t020 - t110)\, \alpha111 + t110\, \alpha121)\, t121 \big)\, \lambda^2$$
$$+ \Bigg(\Bigg(3\, t120\, \alpha111 + 2 \left(h - t010 - t020 + \frac{1}{2}\, t110 - t120 \right) \alpha121 \Bigg)\, t111 - 2\, t121 \Bigg(\Bigg(h$$
$$- t010 - t020 - t110 + \frac{1}{2}\, t120 \Bigg)\, \alpha111 + \frac{3}{2}\, t110\, \alpha121 \Bigg) \Bigg)\, \lambda + \alpha121\, t111^2 + \big((\alpha111$$
$$- \alpha121)\, t121 - 2\, t120\, \alpha111 - \alpha121\, (h - t010 - t020 + t110 - t120) \big)\, t111 + ($$
$$-t121\, \alpha111 + (h - t010 - t020 - t110 + t120)\, \alpha111 + 2\, t110\, \alpha121)\, t121 \Big)\, q^2 + \Big(\big((($$
$$-2\, \alpha111 + 2\, \alpha121)\, t121 + t120\, \alpha111 + \alpha121\, (h - 2\, t010 - 2\, t020 - t120))\, t111$$
$$- ((h - 2\, t010 - 2\, t020 - t110)\, \alpha111 + t110\, \alpha121)\, t121 \big)\, \lambda^2 + \big(((4\, \alpha111$$
$$- 4\, \alpha121)\, t121 - 4\, t120\, \alpha111 - 2\, \alpha121\, (h - 2\, t010 - 2\, t020 + t110 - t120))\, t111$$
$$+ 2\, t121\, ((h - 2\, t010 - 2\, t020 - t110 + t120)\, \alpha111 + 2\, t110\, \alpha121) \big)\, \lambda$$
$$- 2\, \alpha121\, t111^2 + ((-4\, \alpha111 + 4\, \alpha121)\, t121 + 3\, t120\, \alpha111 + \alpha121\, (h - 2\, t010$$

$$- 2\, t020 + 2\, t110 - t120)\,)\, t111 - t121\, (\, -2\, t121\, \alpha l11 + (h - 2\, t010 - 2\, t020 - t110 + 2\, t120)\, \alpha l11 + 3\, t110\, \alpha l21)\,)\, q + \left(-\rho\, t111^2 + \alpha l21\, (t010 + t020)\, t111 - (-\rho\, t121 + \alpha l11\, (t010 + t020)\,)\, t121 \right) \lambda^2 + \left(2\, \rho\, t111^2 + \left(t120\, \alpha l11 - 2\, \alpha l21 \left(t010 + t020 - \frac{1}{2}\, t110 \right) \right) t111 + 2 \left(-\rho\, t121 + \left(t010 + t020 - \frac{1}{2}\, t120 \right) \alpha l11 - \frac{1}{2}\, t110\, \alpha l21 \right) t121 \right) \lambda + (\, \alpha l21 - \rho)\, t111^2 + (\, (\, \alpha l11 - \alpha l21)\, t121 - t120\, \alpha l11 + \alpha l21\, (t010 + t020 - t110)\,)\, t111 - (\, (\, \alpha l11 - \rho)\, t121 + (t010 + t020 - t120)\, \alpha l11 - t110\, \alpha l21)\, t121)\,)$$

$$-\frac{t010\, t020}{\lambda^2} + \frac{P112}{\lambda} - \frac{t111\, t121}{(\lambda - 1)^4} - \frac{t110\, t121 + t111\, t120}{(\lambda - 1)^3} - \frac{-t010\, t020 + tinfty10\, tinfty20 - P112}{(\lambda - 1)^2} - \frac{P112}{\lambda - 1} + \frac{-P112 - a0}{\lambda} + \frac{-t010\, t020 + tinfty10\, tinfty20 - P112 - a2}{(\lambda - 1)^2} + \frac{P112 - a1}{\lambda - 1} - \frac{h\, p}{\lambda - q}$$

```
> rho:=factor(solve(Entry21TermLambdaMinusqCube,rho));
  simplify(rho-(-p*mu));
  simplify(Entry21TermLambdaMinusqCube);
  Lqter:=2*mu*(p-P1(q)/2+h/2/q+2*h/2/(q-1))-h*nu0 -h*nuMinus1*q;
  factor(simplify(Lqbis-Lqter));
```

$$\rho := -\frac{p\, q\, (q - 1)^2\, (\, \alpha l11 - \alpha l21)}{t111 - t121} \tag{2.6}$$

$$0$$

$$0$$

$$Lqter := \frac{1}{t111 - t121} \left(2\, (\, \alpha l11 - \alpha l21)\, q\, (q - 1)^2 \left(p - \frac{1}{2}\, \frac{t010 + t020}{q} - \frac{1}{2}\, \frac{t111 + t121}{(q - 1)^2} - \frac{1}{2}\, \frac{t110 + t120}{q - 1} + \frac{1}{2}\, \frac{h}{q} + \frac{h}{q - 1} \right) \right) - \frac{h\, (\, \alpha l11 - \alpha l21)\, (q - 1)^2}{t111 - t121} - \frac{h\, (\, \alpha l11 - \alpha l21)\, (q - 1)\, q}{t111 - t121}$$

$$0$$

```
> Lp:=simplify(-Entry21TermLambdaMinusq/h):
  Eq5:=simplify(Entry21TermLambdaMinusqSquare-(-p*h*Lq));
```

$$Eq5 := \frac{1}{(q - 1)^2\, (t111 - t121)\, q} \left(6\, h\, (\, \alpha l11 - \alpha l21) \left(\frac{1}{3}\, p^2\, q^6 + \left(-\frac{4}{3}\, p^2 + \left(h - \frac{1}{3}\, t110 - \frac{1}{3}\, t120 - \frac{1}{3}\, t010 - \frac{1}{3}\, t020 \right) p + \frac{1}{3}\, a0 + \frac{1}{3}\, a1 \right) q^5 + \left(2\, p^2 + \left(-\frac{10}{3}\, h - \frac{1}{3}\, t111 + t120 - \frac{1}{3}\, t121 + \frac{4}{3}\, t010 + \frac{4}{3}\, t020 + t110 \right) p + \frac{1}{3}\, t010\, t020 \right. \right. \right.$$ \tag{2.7}

$$-\frac{4}{3}\,a0-a1+\frac{1}{3}\,a2\Big)\,q^4+\Big(-\frac{4}{3}\,p^2+\Big(4\,h+\frac{2}{3}\,t111-t120+\frac{2}{3}\,t121-2\,t010$$
$$-2\,t020-t110\Big)\,p+\frac{1}{3}\,t111\,t120+\frac{1}{3}\,t110\,t121-\frac{4}{3}\,t010\,t020+2\,a0+a1$$
$$-\frac{2}{3}\,a2\Big)\,q^3+\Big(\frac{1}{3}\,p^2+\Big(-2\,h-\frac{1}{3}\,t111+\frac{1}{3}\,t120-\frac{1}{3}\,t121+\frac{4}{3}\,t010+\frac{4}{3}\,t020$$
$$+\frac{1}{3}\,t110\Big)\,p+2\,t010\,t020+\Big(-\frac{1}{3}\,t120+\frac{1}{3}\,t121\Big)\,t111-\frac{1}{3}\,t110\,t121-\frac{4}{3}\,a0$$
$$-\frac{1}{3}\,a1+\frac{1}{3}\,a2\Big)\,q^2+\Big(\Big(\frac{1}{3}\,h-\frac{1}{3}\,t010-\frac{1}{3}\,t020\Big)\,p-\frac{4}{3}\,t010\,t020+\frac{1}{3}\,a0\Big)\,q$$
$$+\frac{1}{3}\,t010\,t020\Big)\Big)$$

```
> a0 := factor(-(p^2*q^6+h*p*q^5-4*p^2*q^5-p*q^5*t010-p*q^5*t020-
  p*q^5*t110-p*q^5*t120-4*h*p*q^4+h*q^4*tinfty10+6*p^2*q^4+4*p*
  q^4*t010+4*p*q^4*t020+3*p*q^4*t110-p*q^4*t111+3*p*q^4*t120-p*
  q^4*t121+q^4*tinfty10*tinfty20+6*h*p*q^3-2*h*q^3*tinfty10-4*
  p^2*q^3-6*p*q^3*t010-6*p*q^3*t020-3*p*q^3*t110+2*p*q^3*t111-3*
  p*q^3*t120+2*p*q^3*t121-2*q^3*t010*t020+q^3*t110*t121+q^3*t111*
  t120-2*q^3*tinfty10*tinfty20-4*h*p*q^2+h*q^2*tinfty10+p^2*
  q^2+4*p*q^2*t010+4*p*q^2*t020+p*q^2*t110-p*q^2*t111+p*q^2*t120-
  p*q^2*t121+5*q^2*t010*t020-q^2*t110*t121-q^2*t111*t120+q^2*
  t111*t121+q^2*tinfty10*tinfty20+h*p*q-p*q*t010-p*q*t020-4*q*
  t010*t020+t010*t020)/(q*(q^2-2*q+1)));
  a1 :=factor( (p^2*q^6+h*p*q^5-4*p^2*q^5-p*q^5*t010-p*q^5*t020-
  p*q^5*t110-p*q^5*t120-4*h*p*q^4+h*q^4*tinfty10+6*p^2*q^4+4*p*
  q^4*t010+4*p*q^4*t020+3*p*q^4*t110-p*q^4*t111+3*p*q^4*t120-p*
  q^4*t121+q^4*tinfty10*tinfty20+5*h*p*q^3-2*h*q^3*tinfty10-4*
  p^2*q^3-6*p*q^3*t010-6*p*q^3*t020-3*p*q^3*t110+2*p*q^3*t111-3*
  p*q^3*t120+2*p*q^3*t121-2*q^3*t010*t020+q^3*t110*t121+q^3*t111*
  t120-2*q^3*tinfty10*tinfty20-2*h*p*q^2+h*q^2*tinfty10+p^2*
  q^2+4*p*q^2*t010+4*p*q^2*t020+p*q^2*t110-p*q^2*t111+p*q^2*t120-
  p*q^2*t121+5*q^2*t010*t020-q^2*t110*t121-q^2*t111*t120+q^2*
  t111*t121+q^2*tinfty10*tinfty20-p*q*t010-p*q*t020-4*q*t010*
  t020+t010*t020)/(q*(q^2-2*q+1)))
  ;
  a2:= factor(-(p^2*q^6+h*p*q^5-4*p^2*q^5-p*q^5*t010-p*q^5*t020-
  p*q^5*t110-p*q^5*t120-3*h*p*q^4+h*q^4*tinfty10+6*p^2*q^4+4*p*
  q^4*t010+4*p*q^4*t020+3*p*q^4*t110-p*q^4*t111+3*p*q^4*t120-p*
  q^4*t121+q^4*tinfty10*tinfty20+3*h*p*q^3-3*h*q^3*tinfty10-4*
  p^2*q^3-6*p*q^3*t010-6*p*q^3*t020-3*p*q^3*t110+2*p*q^3*t111-3*
  p*q^3*t120+2*p*q^3*t121-q^3*t010*t020+q^3*t110*t121+q^3*t111*
  t120-3*q^3*tinfty10*tinfty20-h*p*q^2+3*h*q^2*tinfty10+p^2*
  q^2+4*p*q^2*t010+4*p*q^2*t020+p*q^2*t110-p*q^2*t111+p*q^2*t120-
  p*q^2*t121+3*q^2*t010*t020-q^2*t110*t121-q^2*t111*t120+q^2*
```

```
t111*t121+3*q^2*tinfty10*tinfty20-h*q*tinfty10-p*q*t010-p*q*
t020-3*q*t010*t020-q*tinfty10*tinfty20+t010*t020)/(q*(q^2-2*
q+1)));
simplify(Eq1);
simplify(Eq2);
simplify(Eq5);
```

$$a0 := -\frac{1}{q(q-1)^2}\left(p^2 q^6 + h p q^5 - 4 p^2 q^5 - p q^5 t010 - p q^5 t020 - p q^5 t110 - p q^5 t120 - 4 h p q^4 + h q^4 tinfty10 + 6 p^2 q^4 + 4 p q^4 t010 + 4 p q^4 t020 + 3 p q^4 t110 - p q^4 t111 + 3 p q^4 t120 - p q^4 t121 + q^4 tinfty10\, tinfty20 + 6 h p q^3 - 2 h q^3 tinfty10 - 4 p^2 q^3 - 6 p q^3 t010 - 6 p q^3 t020 - 3 p q^3 t110 + 2 p q^3 t111 - 3 p q^3 t120 + 2 p q^3 t121 - 2 q^3 t010\, t020 + q^3 t110\, t121 + q^3 t111\, t120 - 2 q^3 tinfty10\, tinfty20 - 4 h p q^2 + h q^2 tinfty10 + p^2 q^2 + 4 p q^2 t010 + 4 p q^2 t020 + p q^2 t110 - p q^2 t111 + p q^2 t120 - p q^2 t121 + 5 q^2 t010\, t020 - q^2 t110\, t121 - q^2 t111\, t120 + q^2 t111\, t121 + q^2 tinfty10\, tinfty20 + h p q - p q\, t010 - p q\, t020 - 4 q\, t010\, t020 + t010\, t020\right) \tag{2.8}$$

$$a1 := \frac{1}{q(q-1)^2}\left(p^2 q^6 + h p q^5 - 4 p^2 q^5 - p q^5 t010 - p q^5 t020 - p q^5 t110 - p q^5 t120 - 4 h p q^4 + h q^4 tinfty10 + 6 p^2 q^4 + 4 p q^4 t010 + 4 p q^4 t020 + 3 p q^4 t110 - p q^4 t111 + 3 p q^4 t120 - p q^4 t121 + q^4 tinfty10\, tinfty20 + 5 h p q^3 - 2 h q^3 tinfty10 - 4 p^2 q^3 - 6 p q^3 t010 - 6 p q^3 t020 - 3 p q^3 t110 + 2 p q^3 t111 - 3 p q^3 t120 + 2 p q^3 t121 - 2 q^3 t010\, t020 + q^3 t110\, t121 + q^3 t111\, t120 - 2 q^3 tinfty10\, tinfty20 - 2 h p q^2 + h q^2 tinfty10 + p^2 q^2 + 4 p q^2 t010 + 4 p q^2 t020 + p q^2 t110 - p q^2 t111 + p q^2 t120 - p q^2 t121 + 5 q^2 t010\, t020 - q^2 t110\, t121 - q^2 t111\, t120 + q^2 t111\, t121 + q^2 tinfty10\, tinfty20 - p q\, t010 - p q\, t020 - 4 q\, t010\, t020 + t010\, t020\right)$$

$$a2 := -\frac{1}{q(q-1)^2}\left(p^2 q^6 + h p q^5 - 4 p^2 q^5 - p q^5 t010 - p q^5 t020 - p q^5 t110 - p q^5 t120 - 3 h p q^4 + h q^4 tinfty10 + 6 p^2 q^4 + 4 p q^4 t010 + 4 p q^4 t020 + 3 p q^4 t110 - p q^4 t111 + 3 p q^4 t120 - p q^4 t121 + q^4 tinfty10\, tinfty20 + 3 h p q^3 - 3 h q^3 tinfty10 - 4 p^2 q^3 - 6 p q^3 t010 - 6 p q^3 t020 - 3 p q^3 t110 + 2 p q^3 t111 - 3 p q^3 t120 + 2 p q^3 t121 - q^3 t010\, t020 + q^3 t110\, t121 + q^3 t111\, t120 - 3 q^3 tinfty10\, tinfty20 - h p q^2 + 3 h q^2 tinfty10 + p^2 q^2 + 4 p q^2 t010 + 4 p q^2 t020 + p q^2 t110 - p q^2 t111 + p q^2 t120 - p q^2 t121 + 3 q^2 t010\, t020 - q^2 t110\, t121 - q^2 t111\, t120 + q^2 t111\, t121 + 3 q^2 tinfty10\, tinfty20 - h q\, tinfty10 - p q\, t010 - p q\, t020 - 3 q\, t010\, t020 - q\, tinfty10\, tinfty20 + t010\, t020\right)$$

$$0$$
$$0$$
$$0$$

```
> V:=Matrix(3,3,0):
V[1,1]:=1:
V[1,2]:=1:
V[1,3]:=0:
V[2,1]:=0:
```

```
V[2,2]:=1:
V[2,3]:=1:
V[3,1]:=1/q:
V[3,2]:=1/(q-1):
V[3,3]:=1/(q-1)^2:
V;
B:=Matrix(3,1,0):
B[1,1]:=h*p:
B[2,1]:=h*p*q +t010*t020-tinfty10*(tinfty20+h):
B[3,1]:=p^2-(P1(q)-h*1/q-h*2/(q-1))*p+tdP2(q):
B;
VectorC:=simplify(Multiply(V^(-1),B)):
```

$$\begin{bmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ \frac{1}{q} & \frac{1}{q-1} & \frac{1}{(q-1)^2} \end{bmatrix} \tag{2.9}$$

$$\Big[\big[h\,p\big],\ \big[h\,p\,q + t010\,t020 - (h + tinfty20)\,tinfty10\big],\ \Big[p^2 - \Big(\frac{t010+t020}{q} + \frac{t111+t121}{(q-1)^2} + \frac{t110+t120}{q-1} - \frac{h}{q} - \frac{2\,h}{q-1}\Big)\,p + \frac{t010\,t020}{q^2} + \frac{t111\,t121}{(q-1)^4} + \frac{t110\,t121 + t111\,t120}{(q-1)^3}\Big]\Big]$$

```
> factor(simplify(VectorC[1,1]-C01));
  factor(simplify(VectorC[2,1]-C11));
  factor(simplify(VectorC[3,1]-C12));
```

$$\begin{array}{c} 0 \\ 0 \\ 0 \end{array} \tag{2.10}$$

```
> Lp:=simplify(Lp);
  Lpbis:=mu*(p*diff(P1(q),q)+h*p*(1/q^2+2/(q-1)^2)-diff(tdP2(q),
  q)-C01/q^2-C11/(q-1)^2-2*C12/(q-1)^3)
  +h*nuMinus1*p-h*c11/(q-1)^2:
  factor(simplify(Lp-Lpbis));
```

$$Lp := \frac{1}{(q-1)^3\,(t111-t121)\,q^2}\Big(-3\,p^2\,(\alpha 11 - \alpha 21)\,q^7 - 2\,\Big(h - \frac{13}{2}\,p - t010 - t020 - t110 - t120\Big)\,(\alpha 11 - \alpha 21)\,p\,q^6 - (\alpha 11 - \alpha 21)\,\big(22\,p^2 + (-7\,h + 8\,t010 + 8\,t020 + 7\,t110 - t111 + 7\,t120 - t121)\,p + (h + tinfty20)\,tinfty10\big)\,q^5 + 3\,(\alpha 11 - \alpha 21)\,\big(6\,p^2 + (-3\,h + 4\,t010 + 4\,t020 + 3\,t110 - t111 + 3\,t120 \tag{2.11}$$

$-t121)\, p+(h+tinfty20)\, tinfty10\big)\, q^4+\Big(\big(-7\,\alpha l11+7\,\alpha l21\big)\, p^2+5\,\Big(h-\frac{8}{5}\, t010$
$-\frac{8}{5}\, t020-t110+\frac{3}{5}\, t111-t120+\frac{3}{5}\, t121\Big)\,\big(\alpha l11-\alpha l21\big)\, p+\big(\big(-3\, tinfty10$
$-t121\big)\, h+(t120+t121)\, t111-3\, tinfty10\, tinfty20+t010\, t020+t110\, t121\big)\,\alpha l11$
$+3\,\alpha l21\,\Big(\Big(tinfty10+\frac{1}{3}\, t111\Big)\, h+\Big(-\frac{1}{3}\, t120-\frac{1}{3}\, t121\Big)\, t111+tinfty10\, tinfty20$
$-\frac{1}{3}\, t010\, t020-\frac{1}{3}\, t110\, t121\Big)\Big)\, q^3+\big(p^2\,(\alpha l11-\alpha l21)-(\alpha l11-\alpha l21)\,(h$
$-2\, t010-2\, t020-t110+t111-t120+t121)\, p+\big((tinfty10+t121)\, h+(-t120$
$+t121)\, t111+tinfty10\, tinfty20-3\, t010\, t020-t110\, t121\big)\,\alpha l11-\alpha l21\,\big((tinfty10$
$+t111)\, h+(-t120+t121)\, t111+tinfty10\, tinfty20-3\, t010\, t020-t110\, t121\big)\big)\, q^2$
$+3\, t010\, t020\,(\alpha l11-\alpha l21)\, q-t010\, t020\,(\alpha l11-\alpha l21)\big)$

$$0$$

We have obtained the evolution of the Darboux coordinates

L[q]=2*mu*(p-P1(q)/2+h/2/q+2*h/2/(q-1))-h*nu0 -h*nuMinus1*q

L[p] =mu*(p*diff(P1(q),q)+h*p*(1/q^2+2/(q-1)^2)-diff(tdP2(q),q)-C01/q^2-C11/(q-1)^2-2*C12/(q-1)^3)
+h*nuMinus1*p-h*c11/(q-1)^2

```
> Hamiltonian:= mu*(p^2-P1(q)*p+h*p*(1/q+2/(q-1)) +tdP2(q) )-h*
  nu0*p-h*nuMinus1*q*p -h*c11/(q-1)
  -nuMinus1*(t010*t020-tinfty10*(tinfty20+h)) :
  factor(series(simplify(Lp-(-diff(Hamiltonian,q))),h=0));
  simplify(Lq-(diff(Hamiltonian,p)));
```

$$0 \qquad (2.12)$$
$$0$$

Computation of td{L} and verification of the gauge transformation

```
> simplify(checkL[1,1]);
  checkL11:=-Q2(lambda)/(lambda-0)/(lambda-1)^2;
  simplify(checkL[1,1]-checkL11);
  simplify(checkL[1,2]);
  checkL12:=(lambda-q)/(lambda-0)/(lambda-1)^2;
  simplify(checkL[1,2]-checkL12);
  checkL22bis:=P1(lambda)+Q2(lambda)/(lambda-0)/(lambda-1)^2;
  simplify(checkL[2,2]-checkL22bis);
  checkL21bis:=h*diff( Q2(lambda)/(lambda-q),lambda) +
  L[2,1]*(lambda-0)*(lambda-1)^2/(lambda-q)-P1(lambda)*Q2(lambda)
  /(lambda-q)-Q2(lambda)^2/(lambda-q)/(lambda-0)/(lambda-1)^2:
  simplify(checkL[2,1]-checkL21bis);
```

(3.1)

$$\frac{p\,q\,(q-1)^2}{\lambda\,(\lambda-1)^2} \tag{3.1}$$

$$checkL11 := \frac{p\,q\,(q-1)^2}{\lambda\,(\lambda-1)^2}$$

$$0$$

$$\frac{\lambda-q}{\lambda\,(\lambda-1)^2}$$

$$checkL12 := \frac{\lambda-q}{\lambda\,(\lambda-1)^2}$$

$$0$$

$$checkL22bis := \frac{t010+t020}{\lambda} + \frac{t111+t121}{(\lambda-1)^2} + \frac{t110+t120}{\lambda-1} - \frac{p\,q\,(q-1)^2}{\lambda\,(\lambda-1)^2}$$

$$0$$

$$0$$

```
> G1:=Matrix(2,2,0):
  G1[1,1]:=1:
  G1[1,2]:=0:
  G1[2,1]:=tinfty10*lambda+eta0:
  G1[2,2]:=1:

  dG1dlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dG1dlambda[i,j]:=diff
  (G1[i,j],lambda):  od: od:

  Ltilde:=simplify(Multiply(Multiply(G1,checkL),G1^(-1))+h*
  Multiply(dG1dlambda,G1^(-1))):
  series(Ltilde[1,1],lambda=infinity,2);
  series(Ltilde[1,2],lambda=infinity,2);
  series(Ltilde[2,2],lambda=infinity,2);
  series(Ltilde[2,1],lambda=infinity,2);

  tinfty10+t010+t020+t110+t120-CoherenceEquation1;
  factor(-residue(Ltilde[2,1]/lambda,lambda=infinity)+
  CoherenceEquation1*tinfty10);
  eta0:=factor(solve(factor(-residue(Ltilde[2,1],lambda=infinity)
  ),eta0));
  eta0theo:=1/(tinfty10-tinfty20)*(
  -(2*0*P022+1*P132) +(0^2*C01+1^2*C11+2*1*C12) -h*p*q^2 -
  tinfty10*((t111+t121)+0*(t010+t020)+1*(t110+t120))
  +tinfty10*(tinfty10-tinfty20-h)*(q-1*0-2*1)):
  factor(simplify(eta0-eta0theo+1/(t010+t020+t110+t120+2*
```

```
tinfty10)*CoherenceEquation1*eta0theo));
simplify(-residue(Ltilde[2,1]/lambda,lambda=infinity)+tinfty10*
CoherenceEquation1);
simplify(-residue(Ltilde[2,1],lambda=infinity));
```

$$O\left(\frac{1}{\lambda}\right) \quad (3.2)$$

$$O\left(\frac{1}{\lambda^2}\right)$$

$$O\left(\frac{1}{\lambda}\right)$$

$$O(1)$$

$$-tinfty20$$

$$0$$

$$\eta 0 := \left(p^2 q^6 - 4 p^2 q^5 - p q^5 t010 - p q^5 t020 - p q^5 t110 - p q^5 t120 + 6 p^2 q^4 + 4 p q^4 t010 + 4 p q^4 t020 + 3 p q^4 t110 - p q^4 t111 + 3 p q^4 t120 - p q^4 t121 + q^4 tinfty10^2 - 4 p^2 q^3 - 6 p q^3 t010 - 6 p q^3 t020 - 3 p q^3 t110 + 2 p q^3 t111 - 3 p q^3 t120 + 2 p q^3 t121 - q^3 t110\, tinfty10 - q^3 t111\, tinfty10 - q^3 t120\, tinfty10 - q^3 t121\, tinfty10 - 4 q^3 tinfty10^2 + p^2 q^2 + 4 p q^2 t010 + 4 p q^2 t020 + p q^2 t110 - p q^2 t111 + p q^2 t120 - p q^2 t121 + q^2 t010\, t020 + q^2 t110\, t121 + 2 q^2 t110\, tinfty10 + q^2 t111\, t120 + q^2 t111\, t121 + 2 q^2 t111\, tinfty10 + 2 q^2 t120\, tinfty10 + 2 q^2 t121\, tinfty10 + 5 q^2 tinfty10^2 - p q\, t010 - p q\, t020 - 2 q\, t010\, t020 - q\, t110\, t121 - q\, t110\, tinfty10 - q\, t111\, t120 - q\, t111\, tinfty10 - q\, t120\, tinfty10 - q\, t121\, tinfty10 - 2 q\, tinfty10^2 + t010\, t020\right) \Big/ \left(q\, (q-1)^2 (t010 + t020 + t110 + t120 + 2\, tinfty10)\right)$$

$$0$$

$$0$$

$$0$$

Expression of the Lax matrices in the geometric gauge after the symplectic reduction and the Painlevé 2 equation

Simplification of the formulas after the reduction and expression of the Lax matrices in the geometric gauge after reduction. In this case, we have \check{q}=q and \check{p}=p=\td{p}.

```
> t111:=1/2*t:
alpha111:=1/2:
alpha121:=-1/2:
t121:=-t111:
t020:=-t010:
t120:=-t110:
tinfty20:=-tinfty10:
q:=checkq:
p:=checkp:
simplify(CoherenceEquation1);
```

```
simplify(Trace(L));
simplify(Trace(A));
simplify(Trace(checkL));
simplify(Trace(checkA));
```

$$0 \tag{4.1}$$

$$-\frac{h\left(3\,checkq\,\lambda-2\,\lambda^{2}-checkq\right)}{\lambda\,(\lambda-1)\,(-\lambda+checkq)}$$

$$\frac{(checkq-1)\left(2\,checkp\,checkq^{2}-2\,checkp\,checkq+h\,\lambda\right)}{t\,(-\lambda+checkq)}$$

$$0$$

$$-\frac{h\,(checkq-1)}{t}$$

```
> Ltilde12:=Ltilde[1,2];
factor(simplify(Ltilde[2,2]+Ltilde[1,1]));
eta0bis:=-1/2/tinfty20*( checkq*(checkq-1)^2*checkp^2 -
t010^2/checkq-t^2/4/(checkq-1)^2-t/4*(4*t110+t)/(checkq-1)+
(checkq-2)*tinfty20^2
);
factor(eta0-eta0bis);
```

$$Ltilde12 := \frac{\lambda-checkq}{\lambda\,(\lambda-1)^{2}} \tag{4.2}$$

$$0$$

$$eta0bis := \frac{1}{2}\,\frac{1}{tinfty10}\left(checkq\,(checkq-1)^{2}\,checkp^{2}-\frac{t010^{2}}{checkq}-\frac{1}{4}\,\frac{t^{2}}{(checkq-1)^{2}}-\frac{1}{4}\,\frac{t\,(4\,t110+t)}{checkq-1}+(checkq-2)\,tinfty10^{2}\right)$$

$$0$$

```
> Ltilde111Order2:=factor(simplify(residue(Ltilde[1,1]*
(lambda-1),lambda=1))):
Ltilde111Order1:=factor(simplify(residue(Ltilde[1,1],lambda=1))
):
Ltilde110Order1:=factor(simplify(residue(Ltilde[1,1],lambda=0))
):
Ltilde[1,1]:
Ltilde11bis:= Ltilde111Order2/(lambda-1)^2+Ltilde111Order1/
(lambda-1)+Ltilde110Order1/lambda:
simplify(Ltilde[1,1]-Ltilde11bis);
Ltilde111Order2bis:=(checkq-1)*eta0+(checkq-1)*(checkp*checkq^2
-checkp*checkq+tinfty10):
simplify(Ltilde111Order2-Ltilde111Order2bis);
Ltilde111Order1bis:=-checkq*eta0 -checkp*checkq*(checkq-1)^2-
tinfty10:
```

```
simplify(Ltilde111Order1-Ltilde111Order1bis);
Ltilde110Order1bis:=checkq*eta0+checkp*checkq*(checkq-1)^2:
simplify(Ltilde110Order1-Ltilde110Order1bis);
```

$$0 \quad (4.3)$$
$$0$$
$$0$$
$$0$$

```
> Ltilde211Order2:=factor(simplify(residue(Ltilde[2,1]*
  (lambda-1),lambda=1))):
  Ltilde211Order1:=factor(simplify(residue(Ltilde[2,1],lambda=1))
  ):
  Ltilde210Order1:=factor(simplify(residue(Ltilde[2,1],lambda=0))
  ):
  Ltilde21bis:= Ltilde211Order2/(lambda-1)^2+Ltilde211Order1/
  (lambda-1)+Ltilde210Order1/lambda:
  simplify(Ltilde[2,1]-Ltilde21bis);

  Ltilde211Order2bis:=(checkq-1)*((eta0+checkp*checkq*(checkq-1)+
  tinfty10)^2-(t111^2)/(checkq-1)^2):
  simplify(Ltilde211Order2-Ltilde211Order2bis);
  Ltilde211Order1bis:=-checkq*( (eta0+checkp*(checkq-1)^2)^2 -
  t010^2/checkq^2) :
  simplify(Ltilde211Order1-Ltilde211Order1bis);
  Ltilde210Order1bis:=checkq*( (eta0+checkp*(checkq-1)^2)^2 -
  t010^2/checkq^2):
  simplify(Ltilde210Order1-Ltilde210Order1bis);
```

$$0 \quad (4.4)$$
$$0$$
$$0$$
$$0$$

```
> dqdt:=1/h*Lq:
  dpdt:=simplify(1/h*Lp):
  dcheckqdt:=dqdt;
  dcheckpdt:=dpdt;
  dG1dt:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dG1dt[i,j]:=diff(G1
  [i,j],t)+diff(G1[i,j],checkq)*dcheckqdt+diff(G1[i,j],checkp)
  *dcheckpdt:  od: od:
  Atilde:=simplify(Multiply(Multiply(G1,checkA),G1^(-1))+h*
  Multiply(dG1dt,G1^(-1))):
```

$$dcheckqdt := \frac{2\, checkp\, checkq\, (checkq-1)^2 + h\, checkq^2 - h\, checkq}{h\, t} \quad (4.5)$$

$$dcheckpdt := \frac{1}{4}\, \frac{1}{h\, (checkq-1)^3\, t\, checkq^2} \Big(-12\, checkp^2\, checkq^7 + \big(52\, checkp^2$$

$- 8\, checkp\, h\big)\, checkq^6 + \big(-88\, checkp^2 + 28\, h\, checkp - 4\, (h - tinfty10)\, tinfty10\big)\, checkq^5 + \big(72\, checkp^2 - 36\, h\, checkp + 12\, (h - tinfty10)\, tinfty10\big)\, checkq^4 + \big(-28\, checkp^2 + 20\, checkp\, h - 12\, h\, tinfty10 - t^2 - 4\, t\, t110 - 4\, t010^2 + 12\, tinfty10^2\big)\, checkq^3 + \big(4\, checkp^2 - 4\, checkp\, h + 4\, h\, tinfty10 - t^2 + 4\, t\, t110 + 12\, t010^2 - 4\, tinfty10^2\big)\, checkq^2 - 12\, checkq\, t010^2 + 4\, t010^2\big)$

```
> Atilde12:=simplify(Atilde[1,2]);
simplify(Atilde[1,1]+Atilde[2,2]);
```

$$Atilde12 := \frac{checkq - 1}{(\lambda - 1)\, t} \tag{4.6}$$

$$-\frac{h\, (checkq - 1)}{t}$$

```
> Atilde111Order2:=factor(simplify(residue(Atilde[1,1]*
(lambda-1),lambda=1)));
Atilde111Order1:=factor(simplify(residue(Atilde[1,1],lambda=1))
):
Atilde110Order1:=factor(simplify(residue(Atilde[1,1],lambda=0))
);
Atilde11OrderConstant:=factor(-simplify(residue(Atilde[1,1]
/lambda,lambda=infinity)));
Atilde[1,1]:
Atilde11bis:= Atilde111Order1/(lambda-1)+Atilde11OrderConstant:
simplify(Atilde[1,1]-Atilde11bis);
Atilde11OrderConstantbis:=-tinfty10*(checkq-1)/t:
simplify(Atilde11OrderConstant-Atilde11OrderConstantbis):
Atilde111Order1bis:=-(checkq-1)*(eta0+checkp*checkq^2
-checkp*checkq+tinfty10)/t:
simplify(Atilde111Order1-Atilde111Order1bis);


Atilde221Order2:=factor(simplify(residue(Atilde[2,2]*
(lambda-1),lambda=1)));
Atilde221Order1:=factor(simplify(residue(Atilde[2,2],lambda=1))
):
Atilde220Order1:=factor(simplify(residue(Atilde[2,2],lambda=0))
):
Atilde22OrderConstant:=factor(-simplify(residue(Atilde[2,2]
/lambda,lambda=infinity)));
Atilde22bis:= Atilde221Order1/(lambda-1)+Atilde22OrderConstant:
simplify(Atilde[2,2]-Atilde22bis);
Atilde22OrderConstantbis:=(tinfty10-h)*(checkq-1)/t:
```

```
simplify(Atilde22OrderConstant-Atilde22OrderConstantbis);
Atilde221Order1bis:=(checkq-1)*(eta0+checkp*checkq^2
-checkp*checkq+tinfty10)/t:
simplify(Atilde221Order1-Atilde221Order1bis);
```

$$Atilde111Order2 := 0 \tag{4.7}$$

$$Atilde110Order1 := 0$$

$$Atilde11OrderConstant := -\frac{(checkq-1)\ tinfty10}{t}$$

$$0$$

$$0$$

$$Atilde221Order2 := 0$$

$$Atilde22OrderConstant := -\frac{(checkq-1)\ (h-tinfty10)}{t}$$

$$0$$

$$0$$

$$0$$

```
> Atilde211Order2:=factor(simplify(residue(Atilde[2,1]*
(lambda-1),lambda=1)));
Atilde211Order1:=factor(simplify(residue(Atilde[2,1],lambda=1))
):
Atilde210Order1:=factor(simplify(residue(Atilde[2,1],lambda=0))
);
Atilde21OrderConstant:=factor(-simplify(residue(Atilde[2,1]
/lambda,lambda=infinity)));
Atilde[2,1]:
Atilde21bis:= Atilde211Order1/(lambda-1):
simplify(Atilde[2,1]-Atilde21bis);
Atilde211Order1bis:=-(checkq-1)/t*((eta0+checkp*checkq*
(checkq-1)+tinfty10)^2-t^2/4)-(checkq-2)*checkq*t/(4*(checkq-1)
):
factor(series(simplify(Atilde211Order1-Atilde211Order1bis),p=0)
);
```

$$Atilde211Order2 := 0 \tag{4.8}$$

$$Atilde210Order1 := 0$$

$$Atilde21OrderConstant := 0$$

$$0$$

$$0$$